[CSE301 / Lecture 4]
Side-effects and monads

Noam Zeilberger

Ecole Polytechnique

25 September 2024

What are side-effects?

Everything that a function does besides computing a functional relation from inputs to outputs.

In other words, the difference between “functions in math” and “functions in Python”.

What are side-effects?



versus



“pure” function = no side-effects

```
def sqr(x):
    y = x ** 2
    return y
```

```
>>> sqr(3)
9
>>> sqr(4)
16
```

Printing debugging information

```
def sqr_debug(x):
    y = x ** 2
    print("Squaring {} gives {}!!".format(x, y))
    return y
```

```
>>> sqr_debug(3)
Squaring 3 gives 9!!
9
>>> sqr_debug(4)
Squaring 4 gives 16!!
16
```

Getting input from the user, and raising exceptions

```
def exp_by_input(x):
    k = int(input("Enter exponent: "))
    y = x ** k
    return y
```

```
>>> exp_by_input(3)
Enter exponent: 2
9
>>> exp_by_input(3)
Enter exponent: two
Traceback (most recent call last):
  File "<stdin>", line 1, in <module>
  File "<string>", line 7, in exp_by_input
ValueError: invalid literal for int() with base 10: 'two'
```

Querying a random number generator

```
def exp_by_random(x):
    k = random.randrange(0,10)
    y = x ** k
    return y
```

```
>>> exp_by_random(3)
2187
>>> exp_by_random(3)
243
```

Reading and writing global variables

```
k = 0
def exp_by_counter(x):
    global k
    y = x ** k
    k = k + 1
    return y
```

```
>>> exp_by_counter(3)
1
>>> exp_by_counter(3)
3
>>> exp_by_counter(3)
9
```

Non-standard control flow (e.g., nondeterminism via generators)

```
def exp_by_nondet(x):
    for k in range(0,10):
        y = x ** k
        yield y
```

```
>>> exp_by_nondet(3)
<generator object exp_by_nondet at 0x7ff77d38c830>
>>> list(exp_by_nondet(3))
[1, 3, 9, 27, 81, 243, 729, 2187, 6561, 19683]
```

Side-effects in Haskell

Despite claims, Haskell is not really a pure language...

1. Functions may not terminate
2. Functions may raise exceptions
3. Run-time performance can vary wildly due to laziness

Nevertheless, these effects (at least 1 & 2) are relatively "benign".

In Haskell, most "serious" effects (like getting input from the user, or reading and writing global variables) are confined to *monads.*

The idea, very roughly

Replace



by



where M_{world} captures all possible interactions with the world.

We'll make this more precise, but first let's talk a bit about the principles of *referential transparency* and *compositionality*...

The principle of referential transparency

Informal principle that we can replace an expression by the value it computes without changing the behavior of a program, e.g.:

```
>>> sqr(3)
9
>>> 9 == 9
True
>>> sqr(3) == 9
True
>>> sqr(3) == sqr(3)
True
```

The principle of referential transparency

The presence of side-effects can break referential transparency!

```
>>> exp_by_counter(3)
9
>>> exp_by_counter(3) == 9
False
>>> exp_by_counter(3) == exp_by_counter(3)
False
```

Compositional semantics

More generally, a *semantics* for a programming language is a way of assigning meanings to program expressions. A desired property of a semantics is that it is *compositional*, in the sense that the meaning of an expression is built from the meanings of its subexpressions.

The presence of side-effects presents a challenge to defining a compositional semantics![1] But we can try to surmount it...

[1] Aside: this is also true for semantics of natural languages! See Chung-chieh Shan's PhD thesis, *Lingustic side effects* (2005).

Toy example:[2] arithmetic expressions

Consider a little language of arithmetic expressions, with constants, subtraction, and division:

$$e ::= c \mid e1 - e2 \mid e1 \,/\, e2$$

Each expression e denotes a number $[\![e]\!] \in \mathbb{R}$, defined inductively:

$$[\![c]\!] = c$$
$$[\![e1 - e2]\!] = [\![e1]\!] - [\![e2]\!]$$
$$[\![e1 \,/\, e2]\!] = [\![e1]\!] \,/\, [\![e2]\!]$$

Division by zero is undefined, so $[\![e]\!]$ is sometimes undefined.

[2] Inspired in part by Philip Wadler, "Monads for functional programming", Proceedings of the Båstad Spring School, May 1995.

Toy example: arithmetic expressions

Translated to Haskell:

$$\textbf{data}\ \mathit{Expr} = \mathit{Con}\ \mathit{Double} \mid \mathit{Sub}\ \mathit{Expr}\ \mathit{Expr} \mid \mathit{Div}\ \mathit{Expr}\ \mathit{Expr}$$

$$\begin{aligned}
&\mathit{eval} :: \mathit{Expr} \to \mathit{Double}\\
&\mathit{eval}\ (\mathit{Con}\ c) = c\\
&\mathit{eval}\ (\mathit{Sub}\ e1\ e2) = \mathit{eval}\ e1 - \mathit{eval}\ e2\\
&\mathit{eval}\ (\mathit{Div}\ e1\ e2) = \mathit{eval}\ e1\ /\ \mathit{eval}\ e2
\end{aligned}$$

Toy example: arithmetic expressions

Example expressions:

$$\begin{aligned} e1 &= \mathit{Sub}\ (\mathit{Div}\ (\mathit{Con}\ 2)\ (\mathit{Con}\ 4))\ (\mathit{Con}\ 3) \\ e2 &= \mathit{Sub}\ (\mathit{Con}\ 1)\ (\mathit{Div}\ (\mathit{Con}\ 2)\ (\mathit{Con}\ 2)) \\ e3 &= \mathit{Div}\ (\mathit{Con}\ 1)\ (\mathit{Sub}\ (\mathit{Con}\ 2)\ (\mathit{Con}\ 2)) \end{aligned}$$

And their semantics:

$\mathit{eval}\ e1 = -2.5$ $\mathit{eval}\ e2 = 0$ $\mathit{eval}\ e3$ undefined

Variation #1: error-handling

Modify the semantics to handle division-by-zero.

In a language with exceptions, we could simply raise an exception. Haskell has them, but let's pretend it doesn't and stay "pure"...

Idea: e no longer denotes a number, but a "number or error".

That is, $[\![e]\!] \in \mathbb{R} \uplus \{\mathit{error}\}$

In Haskell, we can return a Maybe type...

Variation #1: error-handling

```
eval1 :: Expr → Maybe Double
eval1 (Con c)    = Just c
eval1 (Sub e1 e2) =
  case (eval1 e1, eval1 e2) of
    (Just x1, Just x2) → Just (x1 − x2)
    _ → Nothing
eval1 (Div e1 e2) =
  case (eval1 e1, eval1 e2) of
    (Just x1, Just x2)
      | x2 ≢ 0 → Just (x1 / x2)
      | otherwise → Nothing
    _           → Nothing
```

Variation #1: error-handling

The example expressions:

$$\begin{aligned} e1 &= \mathit{Sub}\ (\mathit{Div}\ (\mathit{Con}\ 2)\ (\mathit{Con}\ 4))\ (\mathit{Con}\ 3) \\ e2 &= \mathit{Sub}\ (\mathit{Con}\ 1)\ (\mathit{Div}\ (\mathit{Con}\ 2)\ (\mathit{Con}\ 2)) \\ e3 &= \mathit{Div}\ (\mathit{Con}\ 1)\ (\mathit{Sub}\ (\mathit{Con}\ 2)\ (\mathit{Con}\ 2)) \end{aligned}$$

In the new semantics:

$$\mathit{eval1}\ e1 = \mathit{Just}\ (-2.5) \qquad \mathit{eval1}\ e2 = \mathit{Just}\ 0.0$$
$$\mathit{eval1}\ e3 = \mathit{Nothing}$$

Variation #2: global state

Modify the semantics of expressions so that every third constant is interpreted as 0. (Yeah this is a bit weird, but so is most of life.)

The meaning of a subexpression now depends on its position. E.g., $[\![3 - 2]\!] = [\![1]\!]$, but $[\![6 \,/\, (3 - 2)]\!] = [\![6 \,/\, (3 - 0)]\!] \neq [\![6 \,/\, 1]\!]$.

Can we define a compositional semantics?...

Variation #2: global state

...Yes, in state-passing style!

Idea: every subexpression e denotes a function $[\![e]\!] \in \mathbb{N} \to \mathbb{R} \times \mathbb{N}$ taking a count of the previously seen constants, and returning a number together with an updated count.

For the top-level expression, initialize count to 0.

Variation #2: global state

$$
\begin{aligned}
&\mathit{eval2} :: \mathit{Expr} \to \mathit{Int} \to (\mathit{Double}, \mathit{Int}) \\
&\mathit{eval2}\ (\mathit{Con}\ c)\ n = (\textbf{if}\ n\ \text{`}\mathit{mod}\text{`}\ 3 \equiv 2\ \textbf{then}\ 0\ \textbf{else}\ c, n+1) \\
&\mathit{eval2}\ (\mathit{Sub}\ e1\ e2)\ n = \\
&\quad \textbf{let}\ (x1, o) = \mathit{eval2}\ e1\ n\ \textbf{in} \\
&\quad \textbf{let}\ (x2, p) = \mathit{eval2}\ e2\ o\ \textbf{in} \\
&\quad (x1 - x2, p) \\
&\mathit{eval2}\ (\mathit{Div}\ e1\ e2)\ n = \\
&\quad \textbf{let}\ (x1, o) = \mathit{eval2}\ e1\ n\ \textbf{in} \\
&\quad \textbf{let}\ (x2, p) = \mathit{eval2}\ e2\ o\ \textbf{in} \\
&\quad (x1\ /\ x2, p)
\end{aligned}
$$

$$
\begin{aligned}
&\mathit{eval2Top} :: \mathit{Expr} \to \mathit{Double} \\
&\mathit{eval2Top}\ e = \mathit{fst}\ (\mathit{eval2}\ e\ 0)
\end{aligned}
$$

Variation #2: global state

The example expressions:

$$e1 = \mathit{Sub}\ (\mathit{Div}\ (\mathit{Con}\ 2)\ (\mathit{Con}\ 4))\ (\mathit{Con}\ 3)$$
$$e2 = \mathit{Sub}\ (\mathit{Con}\ 1)\ (\mathit{Div}\ (\mathit{Con}\ 2)\ (\mathit{Con}\ 2))$$
$$e3 = \mathit{Div}\ (\mathit{Con}\ 1)\ (\mathit{Sub}\ (\mathit{Con}\ 2)\ (\mathit{Con}\ 2))$$

In the new semantics:

eval2Top e1 $=$ *eval2Top e3* $= 0.5$ *eval2Top e2* undefined

Variation #3: combining error-handling and state

```
eval3 :: Expr → Int → Maybe (Double, Int)
eval3 (Con c) n = Just (if n `mod` 3 ≡ 2 then 0 else c, n + 1)
eval3 (Sub e1 e2) n =
  case eval3 e1 n of
    Nothing → Nothing
    Just (x1, o) → case eval3 e2 o of
      Nothing → Nothing
      Just (x2, p) → Just (x1 − x2, p)
eval3 (Div e1 e2) n =
  case eval3 e1 n of
    Nothing → Nothing
    Just (x1, o) → case eval3 e2 o of
      Nothing → Nothing
      Just (x2, p)
        | x2 ≢ 0 → Just (x1 / x2, p)
        | otherwise → Nothing
```

Variation #3: combining error-handling and state

$$
\begin{aligned}
&\mathit{eval3Top} :: \mathit{Expr} \to \mathit{Maybe\ Double} \\
&\mathit{eval3Top}\ e = \textbf{case}\ \mathit{eval3}\ e\ 0\ \textbf{of} \\
&\quad \mathit{Nothing} \to \mathit{Nothing} \\
&\quad \mathit{Just}\ (x, \_) \to \mathit{Just}\ x
\end{aligned}
$$

In this last semantics:

$$
\begin{gathered}
\mathit{eval3Top}\ e1 = \mathit{eval3Top}\ e3 = \mathit{Just}\ 0.5 \\
\mathit{eval3Top}\ e2 = \mathit{Nothing}
\end{gathered}
$$

Compare with the OCaml version...

```
type expr = Con of float
          | Sub of expr * expr | Div of expr * expr
let cnt = ref 0
let rec eval3 (e : expr) : float =
  match e with
  | Con c -> let n = !cnt in
             (cnt := n+1; if n mod 3 == 2 then 0.0 else c)
  | Sub (e1,e2) -> let x1 = eval3 e1 in
                   let x2 = eval3 e2 in
                   x1 -. x2
  | Div (e1,e2) -> let x1 = eval3 e1 in
                   let x2 = eval3 e2 in
                   if x2 <> 0.0 then x1 /. x2
                   else raise Division_by_zero
let rec eval3Top e = (cnt := 0; eval3 e)
```

Haskell version #3, rewritten using a monad and do notation

```
eval3' :: Expr → StateT Int Maybe Double
eval3' (Con c) = do
  n ← get
  put (n + 1)
  return (if n `mod` 3 ≡ 2 then 0 else c)
eval3' (Sub e1 e2) = do
  x1 ← eval3' e1
  x2 ← eval3' e2
  return (x1 − x2)
eval3' (Div e1 e2) = do
  x1 ← eval3' e1
  x2 ← eval3' e2
  if x2 ≢ 0 then return (x1 / x2) else lift Nothing
eval3' Top e = runStateT (eval3' e) 0 >>= return ∘ fst
```

What is a monad?

A mathematical concept originating in category theory.

Proposed by Eugenio Moggi as a unifying categorical model for different notions of computation.

Adapted by Phil Wadler as a way of integrating side-effects with pure functional programming, in particular in Haskell.

What is a category?

A **category** consists of the following:

- A set of objects, and a set of arrows between objects.
  (Just like a directed graph.)
- For each object a, an identity arrow $a \to a$.
- For each $a \to b$ and $b \to c$, a composite arrow $a \to c$.
- Such that composition and identity are associative and unital.

What is a category?

Examples:

- Set = category whose objects are sets and whose arrows $a \to b$ are *functions* from a to b
- Rel = category whose objects are sets and whose arrows $a \to b$ are *relations* from a to b
- FG = category freely generated from a graph G, with nodes as objects and arrows $a \to b$ given by *paths* from a to b

A category of pure Haskell functions

Informally, we can think of functions of one argument as forming the arrows of a category, whose objects are types.

$$\mathit{Int} \xrightarrow{(+1)} \mathit{Int} \qquad\qquad \mathit{Int} \xrightarrow{(>0)} \mathit{Bool} \qquad\qquad \mathit{etc}.$$

Composition of arrows is defined by function composition.



The identity function $\backslash x \to x$ serves as the identity arrow $a \to a$.

Beyond the category of pure functions

But... we don't always want to program inside this category!

Monads give us a way of building new categories to program in.

A monad is a special kind of *functor.*

What is a functor?

A **functor** is a way of mapping one category into another (possibly the same) category:

- It should map both objects and arrows.
  (Just like a graph homomorphism.)
- It should preserve identity and composition.
  (Just like a monoid/group homomorphism.)

Examples:

- $F\phi : FG \to FH$ where $\phi : G \to H$ is a graph homomorphism
- the powerset functor $P : \mathrm{Set} \to \mathrm{Set}$

Functors in Haskell

The Functor type class:

$$\textbf{class } \mathit{Functor}\ f \textbf{ where}$$
$$\quad \mathit{fmap} :: (a \to b) \to f\ a \to f\ b$$

Note here f is a type constructor $f :: * \longrightarrow *$.

Any instance should satisfy the *functor laws*:

$$\mathit{fmap}\ \mathit{id} = \mathit{id} \qquad \mathit{fmap}\ (f \circ g) = \mathit{fmap}\ f \circ \mathit{fmap}\ g$$

Example #1: the List functor

The list type constructor is a functor:

```
instance Functor [] where
  -- fmap :: (a -> b) -> [a] -> [b]
  fmap = map
```

(Exercise: prove the functor laws $map\ id\ xs = xs$ and $map\ (f \circ g)\ xs = map\ f\ (map\ g\ xs)$ by structural induction!)

Example #2: the Maybe functor

The *Maybe* type constructor is a functor:

```
instance Functor Maybe where
  -- fmap :: (a -> b) -> Maybe a -> Maybe b
  fmap f Nothing = Nothing
  fmap f (Just x) = Just (f x)
```

Rough definition of a monad, in category theory

A **monad** is a functor m from a category to itself, equipped with an arrow $a \to m\,a$ for every object a, together with a way of transforming arrows $a \to m\,b$ into arrows $m\,a \to m\,b$, subject to certain equations.

For example, the powerset functor is a monad: the functions $a \to m\,a$ are defined by taking singletons, and any function $a \to m\,b$ extends to a function $m\,a \to m\,b$ by taking unions.

Rough definition of a monad, in category theory

What's important about the definition is that it allows to build a new category with the same objects, but where an arrow $a \to b$ in the new category corresponds to an arrow $a \to m\, b$ in the old category. This is called the "Kleisli category" construction.

For example, taking the Kleisli category of the powerset monad gives a category with sets as objects but whose arrows are *relations*.

Monads in Haskell (before 2014)

The Monad type class:

$$\textbf{class } \mathit{Monad}\ m \textbf{ where}$$
$$\quad \mathit{return} :: a \to m\ a$$
$$\quad (\gg\!=) :: m\ a \to (a \to m\ b) \to m\ b \quad \text{-- pronounced "bind"}$$

Subject to the *monad laws*:

$$\mathit{return}\ x \gg\!= f = f\ x$$
$$mx \gg\!= \mathit{return} = mx$$
$$(mx \gg\!= f) \gg\!= g = mx \gg\!= (\backslash x \to (f\ x \gg\!= g))$$

(Note that $\mathit{flip}\ (\gg\!=) :: (a \to m\ b) \to (m\ a \to m\ b)$.)

The List and Maybe monads

```
instance Monad [] where
  -- return :: a -> [a]
  return x = [x]
  -- (>>=) :: [a] -> (a -> [b]) -> [b]
  xs >>= f = concatMap f xs
```

```
instance Monad Maybe where
  -- return :: a -> Maybe a
  return x = Just x
  -- (>>=) :: Maybe a -> (a -> Maybe b) -> Maybe b
  Nothing >>= f = Nothing
  Just x >>= f = f x
```

Monads as notions of computation

Via the Kleisli category constructions...

- List monad: category of nondeterministic functions $a \to [b]$.
- Maybe monad: category of partial functions $a \to \mathit{Maybe}\ b$.

Evaluator #1, re-expressed using the Maybe monad

$$
\begin{aligned}
&\mathit{eval1'} :: \mathit{Expr} \to \mathit{Maybe\ Double} \\
&\mathit{eval1'}\ (\mathit{Con\ c}) = \mathit{return\ c} \\
&\mathit{eval1'}\ (\mathit{Sub\ e1\ e2}) = \\
&\quad \mathit{eval1'\ e1} \gg\!= \backslash \mathit{x1} \to \\
&\quad \mathit{eval1'\ e2} \gg\!= \backslash \mathit{x2} \to \\
&\quad \mathit{return}\ (\mathit{x1} - \mathit{x2}) \\
&\mathit{eval1'}\ (\mathit{Div\ e1\ e2}) = \\
&\quad \mathit{eval1'\ e1} \gg\!= \backslash \mathit{x1} \to \\
&\quad \mathit{eval1'\ e2} \gg\!= \backslash \mathit{x2} \to \\
&\quad \textbf{if}\ \mathit{x2} \not\equiv 0\ \textbf{then}\ \mathit{return}\ (\mathit{x1}\ /\ \mathit{x2})\ \textbf{else}\ \mathit{Nothing}
\end{aligned}
$$

The State monad

The type constructor *State s* is defined essentially as follows:

```
newtype State s a = State { runState :: s -> (a, s) }
```

It is a monad:

```
instance Monad State where
  return x = State (\s -> (x, s))
  xt >>= f = State (\s0 ->
    let (x, s1) = runState xt s0 in
    runState (f x) s1)
```

The State monad

Also, it supports "get" and "set" operations:

$$
\begin{aligned}
&\mathit{get} :: \mathit{State}\ s\ s \\
&\mathit{get} = \mathit{State}\ (\backslash s \rightarrow (s, s)) \\
&\mathit{put} :: s \rightarrow \mathit{State}\ s\ () \\
&\mathit{put}\ s' = \mathit{State}\ (\backslash s \rightarrow ((), s'))
\end{aligned}
$$

Evaluator #2, redefined using the State monad

$$
\begin{aligned}
&\mathit{eval2'} :: \mathit{Expr} \to \mathit{State\ Int\ Double} \\
&\mathit{eval2'}\ (\mathit{Con}\ c) = \\
&\quad \mathit{get} \gg\!= \backslash n \to \\
&\quad \mathit{put}\ (n + 1) \gg\!= \backslash \_ \to \\
&\quad \mathit{return}\ (\textbf{if}\ n\ `\mathit{mod}`\ 3 \equiv 2\ \textbf{then}\ 0\ \textbf{else}\ c) \\
&\mathit{eval2'}\ (\mathit{Sub}\ e1\ e2) = \\
&\quad \mathit{eval2'}\ e1 \gg\!= \backslash x1 \to \\
&\quad \mathit{eval2'}\ e2 \gg\!= \backslash x2 \to \\
&\quad \mathit{return}\ (x1 - x2) \\
&\mathit{eval2'}\ (\mathit{Div}\ e1\ e2) = \\
&\quad \mathit{eval2'}\ e1 \gg\!= \backslash x1 \to \\
&\quad \mathit{eval2'}\ e2 \gg\!= \backslash x2 \to \\
&\quad \mathit{return}\ (x1\ /\ x2) \\
&\mathit{eval2Top'}\ e = \mathit{fst}\ (\mathit{runState}\ (\mathit{eval2'}\ e)\ 0)
\end{aligned}
$$

Do notation

$$
\begin{aligned}
&\textbf{do } x1 \leftarrow e1 \\
&\quad x2 \leftarrow e2 \\
&\quad \ldots \\
&\quad xn \leftarrow en \\
&\quad f\ x1\ x2\ \ldots\ xn
\end{aligned}
$$

is syntactic sugar for

$$
\begin{aligned}
&e1 \gg\!= \backslash x1 \rightarrow \\
&e2 \gg\!= \backslash x2 \rightarrow \\
&\ldots \\
&en \gg\!= \backslash xn \rightarrow \\
&f\ x1\ x2\ \ldots\ xn
\end{aligned}
$$

Evaluator #2, equivalently expressed with do notation

$$
\begin{aligned}
&\mathit{eval2'} :: \mathit{Expr} \to \mathit{State\ Int\ Double} \\
&\mathit{eval2'}\ (\mathit{Con}\ c) = \textbf{do} \\
&\quad n \leftarrow \mathit{get} \\
&\quad \mathit{put}\ (n + 1) \\
&\quad \mathit{return}\ (\textbf{if}\ n\ \text{`}\mathit{mod}\text{`}\ 3 \equiv 2\ \textbf{then}\ 0\ \textbf{else}\ c) \\
&\mathit{eval2'}\ (\mathit{Sub}\ e1\ e2) = \textbf{do} \\
&\quad x1 \leftarrow \mathit{eval2'}\ e1 \\
&\quad x2 \leftarrow \mathit{eval2'}\ e2 \\
&\quad \mathit{return}\ (x1 - x2) \\
&\mathit{eval2'}\ (\mathit{Div}\ e1\ e2) = \\
&\quad x1 \leftarrow \mathit{eval2'}\ e1 \\
&\quad x2 \leftarrow \mathit{eval2'}\ e2 \\
&\quad \mathit{return}\ (x1\ /\ x2)
\end{aligned}
$$

The IO monad

A built-in monad used to perform real system I/O.

Supports operations like

$$\begin{aligned} &\mathit{getLine} :: \mathit{IO\ String} \\ &\mathit{putStrLn} :: \mathit{String} \to \mathit{IO}\ () \end{aligned}$$

etc.

The use of a monad ensures proper sequentialization, as we can never "escape" the IO monad![3]

[3] Technically, this is not true. There is a back door in the form of a function $\mathit{unsafePerformIO} :: \mathit{IO}\ a \to a$, contained in the module *System.IO.Unsafe*. But as the name suggests, this function should be used with care...

Monads in Haskell (post 2014)

A bit more heavy since the "Functor-Applicative-Monad" hierarchy:

$$\textbf{class } \mathit{Functor}\ f \textbf{ where}$$
$$\quad \mathit{fmap} :: (a \to b) \to f\ a \to f\ b$$
$$\textbf{class } \mathit{Functor}\ f \Rightarrow \mathit{Applicative}\ f \textbf{ where}$$
$$\quad \mathit{pure} :: a \to f\ a$$
$$\quad (\langle * \rangle) :: f\ (a \to b) \to f\ a \to f\ b$$
$$\textbf{class } \mathit{Applicative}\ m \Rightarrow \mathit{Monad}\ m \textbf{ where}$$
$$\quad \mathit{return} :: a \to m\ a$$
$$\quad (\gg\!=) :: m\ a \to (a \to m\ b) \to m\ b$$
$$\quad \mathit{return} = \mathit{pure}$$

So to define an instance of Monad, you first need instances of Functor and Applicative.

Monads in Haskell (post 2014)

But instances of Functor and Applicative can always be retrofitted from a Monad instance:

```
instance Functor M where
  fmap f xm = xm >>= return . f
instance Applicative M where
  pure = return
  fm <*> xm = fm >>= \f -> xm >>= return . f
```

Combining monads

To define Evaluator #3, we implicitly used a *monad transformer*:

$$\textbf{newtype}\ \mathit{StateT}\ s\ m\ a = \mathit{StateT}\ \{\mathit{runStateT} :: s \to m\ (a, s)\}$$

Given a monad m representing some notion of computation (e.g., partiality or nondeterminism), $\mathit{StateT}\ s\ m$ defines a new monad with s state wrapped around an m-computation.

But it is not always clear how to combine monads.

More generally, the question of how to organize and reason about programs with side-effects remains an important open problem!